

RECIP_PARETE_SOTTILE_She11208

```

FINISH
/CLEAR
C*****
C***
C*** RECIPIENTE IN PRESSIONE IN PARETE SOTTILE
C***
C*** ELEMENTI GUSCIO ASSIALSIMMETRICO
C***
C*** PARAMETRI
/PREP7
D=1800      ! diametro recipiente
L=4000      ! lunghezza recipiente (fondi esclusi)
HF=400      ! semi-asse minore fondi
SP=3        ! spessore
PINT=1      ! pressione interna
NC=50       ! numero elementi fasciame cilindrico
NR=10       ! numero elementi zona raccordo
NF=40       ! numero elementi fondo
RE=4        ! rapporto dimensioni elementi
RSP=5       ! dimensioni zona infittita giunzione/spessore
C***
C*** NODI
C***
N,1,D/2
N,NC+1,D/2,L/2-SP*RSP
FILL,1,NC+1,,,,,1/RE
NGEN,NR+1,1,NC+1,NC+1,,0,SP*RSP/NR
LOCAL,11,1,0,L/2,0,,,,HF/(D/2)
NGEN,NR+1,1,NC+NR+1,,,0,((SP*RSP)/HF/3.14*180)/NR
N,NC+2*NR+NF+1,D/2,90
FILL,NC+2*NR+1,NC+2*NR+NF+1,,,,,RE
CSYS,0
C***
C*** MATERIALE
C***
MP,EX,1,210000
MP,NUXY,1,0.3
MP,GXY,1,76000
C***
C*** ELEMENTI
C***
ET,1,208
SECTYPE,1,SHELL
SECDATA,SP
E,1,2
EGEN,NC+2*NR+NF,1,-1
C***
C*** VINCOLI
C***
D,1,UY,0
D,1,ROTZ,0
D,1,UZ,0
D,NC+2*NR+NF+1,UX,0
D,NC+2*NR+NF+1,ROTZ,0
C***
C*** CARICHI
C***
SFE,ALL,1,PRES,,PINT
FINISH
C***
C*** SOLUZIONE
C***
/SOLU
SOLVE
FINISH
C***
C*** POST-PROCESSING
C***

```

RECIP_PARETE_SOTTILE_She11208

```

/POST1
C*** opzioni di rappresentazione
/PBC,ALL,,1
/PBC,NFOR,,0
/PBC,NMOM,,0
/PBC,RFOR,,0
/PBC,PATH,,0
/PNUM,NODE,0
/PSF,PRES,NORM,2
C*** struttura con vincoli e carichi
/TITLE,MODELLO
EPLOT
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

C*** deformata
/PBC,ALL,,0
/PSF,PRESS,NORM,0
/TITLE,DEFORMATA
PLDISP,1

*ASK,IFL,PREMERE RETURN PER CONTINUARE,0
C*** tensioni
/PBC,ALL,,0
ETABLE,SLM,SMISC,18      ! estrae il dato "tensione membranale longitudinale"
ETABLE,SCM,SMISC,19      ! estrae il dato "tensione membranale
circonferenziale"
ETABLE,SLF,SMISC,21      ! estrae il dato "tensione flessionale
longitudinale"
ETABLE,SCF,SMISC,22      ! estrae il dato "tensione flessionale
circonferenziale"
ETABLE,STM,SMISC,29      ! estrae il dato "tensione tangenziale trasversale"

/TITLE,TENSIONE LONGITUDINALE (MEMBRANALE)
PLLS,SLM,SLM
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

/TITLE,TENSIONE LONGITUDINALE (FLESSIONALE)
PLLS,SLF,SLF
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

/TITLE,TENSIONE CIRCONFERENZIALE (MEMBRANALE)
PLLS,SCM,SCM
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

/TITLE,TENSIONE CIRCONFERENZIALE (FLESSIONALE)
PLLS,SCF,SCF
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

/TITLE,TENSIONE DI TAGLIO NELLO SPESSORE
PLLS,STM,STM
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

```